

方法介绍

岩组资料的数据处理及成图 自动化 (之一)

——岩组图的制作

李福佩
(浙江大学地质系)

吴乃爵
(浙江工学院)

岩组资料的数据处理实质上是数理统计方法在岩组分析中的应用。它通过对样本——一个个晶体的光轴或解理面——空间方位的测定,以概率统计方法来了解和判断岩石总体的统计特性。既要应用数理统计这一手段,就要求有一定数量的样本,而对大量样本的统计,必然导致岩组数据处理工作之浩繁。本文是笔者以“APPLE-II”为工具,将繁杂的数据处理以及岩组的成图工作交给电脑去完成的一份总结。它包括展点、统计、等密图的绘制、岩组图的旋转以及岩组文本文件和图形文件的制备。

一、展点:把观测的各个光轴或解理面的方位(方位角、倾向、倾角)依极射赤平投影的方法投影在施密特网上。

基本构思:手工展点是在施密特网不动的情况下,旋转作图纸进行操作,而电脑展点则利用电脑能快速运算的特点,在确定起始箭头方位的情况下,通过函数运算,精确地定出不同方位角,不同倾向、倾角时光轴出露点或解理面极点在施密特网上的坐标。

基本原理:依赤平投影的原理,任一确定的方位角 N ,可以按不同的倾向、倾角投影在东、西、南、北径上(下半球投影)。如某晶体要素的测量结果为:方位角 $N=50^\circ$,倾角 J 为 $10^\circ E$,则投影位置如图1所示。




图 1




图 2

1—旋转前施氏网坐标轴线, 2—旋转后施氏网坐标轴线

我们让箭头始终位于圆的正下方,让施密特网按方位角的大小逆时针旋转以保持方位角与投影线之间的相对位置(图2)。

其次依倾角大小确定投影点在投影线上的位置。鉴于极射赤平投影中H环向东、西倾光

轴躺平时, 倾角投影点从大圆至圆心为 $0^\circ \sim 90^\circ$, 而弗氏台向南、北倾光轴立直时, 其倾角投影点从大圆至圆心为 $90^\circ \sim 0^\circ$, 我们可以把展点工作归纳为求算光轴出露点(或解理面极点)在赤平投影网上的极坐标问题:

方位角 $N = 450^\circ - N$ (向东倾)

$N = 630^\circ - N$ (向西倾)

$N = 360^\circ - N$ (向南倾)

$N = 540^\circ - N$ (向北倾)

离圆心的距离 $D = 2RASIN\left(\frac{90 - J}{2}\right)$ (向东、西倾)

$D = 2RASIN\frac{J}{2}$ (向南、北倾)

其中 R 为施密特网半径, A 为弦长变为施密特网投影的比例系数, J 为倾角。

电脑的工作就是把每个光轴(或解理面)的测量数据变为它们相应的极坐标(N, D), 并将其标示出来。

若化为直角坐标:

$$X = X_0 - DSIN(N)$$

$$Y = Y_0 + DCOS(N)$$

(X_0, Y_0)为施密特网圆心的坐标, N 为方位角。

于是每个光轴出露点就有与之对应的(X, Y)坐标, 若测量100个矿物颗粒, 则可安排两个单下标变量 $M(100)$ 及 $N(100)$ 分别存储100个点的 X 坐标及 Y 坐标。

电脑展点时:

```
70 HGR:HCOLOR=7
```

```
120 P=P+1 计数器, 记录测量点数
```

```
130 INPUT N, J, I 从键盘输入测量的方位角、倾向、倾角
```

```
260 M(P)=70-D*SIN(N):N(P)=70+D*COS(N):HPLOTM(P), N(P) 经换算后
```

求出该点的直角坐标及展点

```
270 IF P=M GOTO 290 M为测量总点数
```

```
280 GOTO 120
```

二、统计: 展点后如图3所示:

由于APPLE-II电脑在高分辨作图情况下坐标 X 为 $0 \sim 279$, Y 为 $0 \sim 159$, 故将施密特网圆心置于 $(70, 70)$, 并令其半径 R 也等于70。同时设计一个 7×7 的方格网复盖于施氏网之上, 这样每个节点周围的四个方格就可以对应于半径为1cm的小圆孔方板进行统计, 它的面积相当于 4cm^2 , 比方板上小圆孔面积 $\pi r^2 = 3.14159\text{cm}^2$ 大, 统计时将使极点数偏高, 但全




图3

图皆系统偏高,并不影响等密图的形状及它们之间的相对位置。我们以方格网上的每个节点为中心,统计围绕节点附近四个方格中的极点数,并将此数赋于该节点。如:

点(0,0)相当于X(1);点(7,0)相当于X(2);...点(140,0)相当于X(21);

点(0,7)相当于X(22);点(7,7)相当于X(23);...点(140,140)相当于X(441)。

统计程序是这样安排的:

```

320 FOR I=0 TO 140 STEP 7 以I为Y轴
340 FOR J=0 TO 140 STEP 7 以J为X轴
350 FOR K=1 TO 100 K为测量点
360 IF M(K)>J-7 AND M(K)<J+7 AND (NK)>I-7 AND N(K)<I+7
THEN S=S+1
370 NEXT K
350~370: 100个测量点中有几个点落在J-7, J+7, I-7, I+7这一区间里。
380 E=E+1:X(E)=S:S=0 将统计结果赋于X(E)节点
390 NEXT J:NEXT I

```

而对大圆附近的统计是求出量板在直径两端两个半圆统计数之和,故需依大圆上的节点统计数进行一系列的求和及重复赋值。

如A节点为X(36),B点节点为X(406),其相应的面积之和为大对四个方格(见图3),则应除去其增大的部分。故为:

```
420 X(36)=INT(((X(36)+X(406))/1.125):X(406)=X(36)
```

三、岩组图的绘制:

在统计了施氏网的7×7网格上各节点的极点数后,只要用线性内插法将节点与节点之间的各个整数部分求出来,并将等值点联系在一起就完成了等密图的绘制。以二维数组W(X, Y)替换X(I),则二维数组不但表示了各节点处统计后的极点数,而且还给出了该点的坐标(X, Y)。

典型的内插程序为:

```

810 FOR Y=0 TO 7
820 FOR X=0 TO 133 STEP 7
830 IF W(X,Y)=W(X+7,Y) GOTO 880
840 A1=ABS(W(X,Y)-W(X+7,Y)):A2=INT(A1/7*100+0.5)/100
850 IF W(X,Y)>W(X+7,Y) GOTO 870
860 FOR I=1 TO 6:W(X+I,Y)=W(X,Y)+A2*I:NEXT I:GOTO 890
870 FOR I=1 TO 6:W(X+7-I,Y)=W(X+7,Y)+A2*I:NEXT I:GOTO
890
880 FOR I=1 TO 6:W(X+I,Y)=W(X,Y):NEXT I
890 NEXT X:NEXT Y

```

这是沿X方向逐行内插的程序,实质上是沿着X方向及Y方向安排的二重嵌套循环。

然后采用逐点内插逐行扫描的办法来完成等密图的绘制。但必须考虑到内插时各坐标点得到的极点数多为非整数,故显示时使 $I=I\pm 0.2$ 来显示。通过逐行扫描,则可得出一帧完整

的等值线图。

四、图件对比：

1. 人工绘制：

2. 电脑绘制：




图 4




图 5

五、程序使用及说明：可依程序中提供的注释操作。兹分段说明如下：

作为某矿区某块标本测定的原始数据可以经命名后将展点时各测点的坐标做成文本文件存于磁盘之中。30句为用户提供了找寻某标本的原始资料咨询。

同样，该标本的岩组图也可以作为图形文件将图形存于磁盘中。50句提供了索检图形的可能。

70~90：绘制大圆及箭头。

110~120：输入测量的颗粒数及颗粒计数。

130：输入测量的方位角(N)，倾角(J)，倾向(I)。

140~250：将N, J, I换算为极坐标：N(方位角)，D(圆心距)。

260：求算该光轴出露点的X坐标M(P)及Y坐标N(P)，并展点。

290~320：提供了将展点图件打印出来的功能。

330~490：为统计程序。将统计结果赋值于X(1)~X(441)节点，并将统计结果显示于屏幕。

550~560：可将统计结果作为文本文件存储于磁盘的程序。

570~640：这是为今后欲调用文本文件时预设的。它可将文本文件从磁盘输入给电脑。

650~1020：这是重复赋值、分区作图的程序。沿Y方向分为Q=0~19共20个作图区。

其中：

-670~710随Q的变化重复赋值。

720~800: 为节点之间沿Y方向内插。

810~890: 为节点之间沿X方向内插。

从而求出各个坐标点所拥有的节点数。

900~980, 1130~1180: 为逐行扫描绘制等密图的程序。

1030~1070: 为抹去大圆以外不需要的图形并将大圆及箭头重新绘出。

1080~1110: 可将绘制好的岩组图作为图形文件存入磁盘, 以备将来调用。

1190~1240: 为调用图形文件及打印图形的程序。

参 考 文 献

- [1] 何作霖, 1965, 赤平极射投影在地质科学中的应用, 科学出版社。
- [2] 何绍勋, 1979, 构造地质学中的赤平极射投影, 地质出版社。
- [3] 张巨洪等, 1983, BASIC语言程序库, 清华大学出版社。
- [4] Loudon, T.V., 1979, Computer Methods in Geology.

程序清单

```

10 REM DATA PROCESSING OF PETROFABRIC DIAGRAM
20 LOMEM:16384: DIM X(441), W(140,7), M(100), N(100)
30 PRINT "DO YOU WANT TO RETRIEVE THE TEXT FILE TO DRAW A DIAG-
RAM? (ANS. Y OR N)": INPUT G$
40 IF G$ = "Y" GOTO 500
50 PRINT "DO YOU WANT TO RETRIEVE THE GRAPHIC FILE? (ANS. Y OR N)":
INPUT G$
60 IF G$ = "Y" GOTO 1190
70 HGR:HCOLOR=7:R=70:G=3.14159/180
80 FOR TH=0 TO 360: RH=TH*G: X=70+R*COS(RH): Y=70-R*SIN(RH):
HPlot X, Y: NEXT
90 HPlot 68, 144 TO 70, 142 TO 70, 147
100 VTAB 22
110 INPUT "DETERMINATION NUMBER =", M
120 P=P+1
130 INPUT "BEARING =", N:INPUT "DIP ANGLE =", J: INPUT "I = (E=1, W=2,
S=3, N=4)", I
132 IF N>360 OR N<0 OR J>90 OR J<0 OR I4> OR I<0 GOTO 1250
140 IF I=1 GOTO 180
150 IF I=2 GOTO 190
160 IF I=3 GOTO 200
170 IF I=4 GOTO 210
180 N=450-N:GOTO 220
190 N=630-N:GOTO 220
200 N=360-N:GOTO 240
210 N=540-N:GOTO 240
220 J=(90-J)*G

```

```

230 GOTO 250
240 J = J * G
250 D = 2 * R / SQR(2) * SIN(J/2) : N = N * G
260 M(P) = 70 - D * SIN(N) : N(P) = 70 + D * COS(N) : HPLOT M(P), N(P)
270 IF P = M GOTO 282
280 GOTO 120
282 PPINT " * REVISE * : PRESS HCOLOR = O : GOTO 130 : PRESS WRONG DATA THEN :
    HCOLOR = 7 : GOTO 130 : PRESS CORRECT DATA. AT LAST PRESS 'CONT' "
284 STOP
290 PRINT "DO YOU WANT TO PRINT THE DIAGRAM? (ANS. Y OR N)"
300 INPUT A$
310 IF A$ = "N" GOTO 330
320 RR 1 : PRINT : PRINT TAB(8) "POINT-GROUP DIAGRAM" : POKE 1913, 65 : PRINT
    CHR$(17)
322 PRINT TAB(10); "** WAIT PLEASE**"
325 E = 0 : S = 0
330 FOR I = 0 TO 140 STEP 7
340 FOR J = 0 TO 140 STEP 7
350 FOR K = 1 TO 100
360 IF M(K) > J - 7 AND M(K) < J + 7 AND N(K) > J - 7 AND N(K) < I + 7 THEN S = S + 1
370 NEXT K
380 E = E + 1 : X(E) = S : S = 0
390 NEXT J : NEXT I
400 X(11) = X(11) + X(431) : X(431) = X(11) : X(12) = X(12) + X(430) : X(430) = X(12)
    : X(13) = X(13) + X(429) : X(429) = X(13) : X(10) = X(10) + X(432) : X(432) = X
    (10) : X(9) = X(9) + X(433) : X(433) = X(9)
410 X(413) = X(413) + X(8) : X(8) = X(413) : X(407) = X(14) + X(407) : X(14) = X
    (407) : X(35) = X(428) + X(35) : X(482) = X(35) : X(29) = X(29) + X(434) : X(434)
    = X(29) : X(28) = INT((X(28) + X(414)) / 1.125) : X(414) = X(28)
420 X(36) = INT((X(36) + X(406)) / 1.125) : X(406) = X(36) : X(231) = X(231) + X(211) :
    X(211) = X(231) : X(190) = X(190) + X(252) : X(252) = X(190) : X(169) = X(169) +
    X(273) : X(273) = X(169) : X(210) = X(210) + X(232)
430 X(232) = X(210) : X(189) = X(189) + X(253) : X(253) = X(189) : X(275) = X(275) +
    X(168) : X(168) = X(275) : X(149) = X(149) + X(294) : X(294) = X(149) : X(293)
    = X(293) + X(148) : X(148) = X(293)
440 X(167) = X(167) + X(274) : X(274) = X(167) : X(296) = INT((X(296) + X(146)) /
    1.125) : X(146) = X(296) : X(128) = INT((X(128) + X(314)) / 1.1) : X(314) = X(128) :
    X(48) = INT((X(48) + X(394)) / 1.625) : X(394) = X(48)
450 X(58) = INT((X(58) + X(384)) / 1.625) : X(384) = X(58) : X(59) = X(59) + X(383) :
    X(383) = X(59) : X(47) = X(47) + X(395) : X(395) = X(47) : X(67) = X(67) + X
    (375) : X(375) = X(67)

```

```

460 X(87) = X(87) + X(355) : X(355) = X(87) : X(81) = X(81) + X(361) : X(361) = X
(81) : X(103) = X(103) + X(339) : X(339) = X(103) : X(68) = X(68) + X(396) : X
(396) = X(68) : X(80) = X(80) + X(382) : X(382) = X(80)
470 X(102) = X(102) + X(360) : X(360) = X(102) : X(124) = X(124) + X(338) : X(338) =
X(124) : X(88) = X(88) + X(376) : X(376) = X(88) : X(108) = X(108) + X(356) : X
(356) = X(108) : X(374) = X(374) + X(46)
480 X(46) = X(374) : X(362) = X(362) + X(60) : X(60) = X(362) : X(354) = X(354) + X
(66) : X(66) = X(354) : X(340) = X(340) + X(82) : X(82) = X(340) : X(334) = X(334)
+ X(86) : X(86) = X(334) : X(318) = X(318) + X(104) : X(104) = X(318)
490 FOR I=1 TO 441:PRINT " ",X(I);:NEXT:PRINT O:PRINT
500 PRINT"DO YOU WANT TO SET UP A TEXT FILE? (ANS. Y OR N)"
510 INPUT A$:D$ =CHR$(4)
520 IF A$ = "N"GOTO 570
530 INPUT"WHAT'S THE FILE'S NAME? ",N$
540 PRINT D$,"MON C,O":PRINT D$,"OPEN";N$:PRINT D$,"WRITE";N$
550 FOR I=1 TO 441:PRINT X(I):NEXT
560 PRINT D$,"CLOSE";N$
570 PRINT"DO YOU WANT TO RETRIEVE THE TEXT FILE?(ANS. Y OR N)"
580 INPUT B$:D$ =CHR$(4)
590 IF B$ = "N"GOTO 650
600 INPUT"NAME OF THE TEXT FILE?";Z$
610 PRINT D$,"MON C,I,O"
620 PRINT D$,"OPEN";Z$:PRINT D$,"READ";Z$
630 FOR I=1 TO 441:INPUT X(I):NEXT
640 PRINT D$,"CLOSE";Z$
650 HGR:HCOLOR=7
660 FOR Q=0 TO 19
670 FOR Y=0 TO 7 STEP 7
680 FOR X=0 TO 140 STEP 7
690 A=A+1:W(X,Y)=X(A)
700 PRINT " ",X(A),
710 NEXT X:NEXT Y
720 FOR X=0 TO 140 STEP 7
730 Y=0
740 IF W(X,Y)=W(X,Y+7)GOTO 790
750 A1=ABS(W(X,Y)-W(X,Y+7)):A2=INT(A1/7*100+0.5)/100
760 IF W(X,Y)>W(X,Y+7)GOTO 780
770 FOR I=1 TO 6:W(X,Y+I)=W(X,Y)+I*A2:NEXT I:GOTO 800
780 FOR I=1 TO 6:W(X,Y+7-I)=W(X,Y+7)+I*A2:NEXT I:GOTO 800
790 FOR I=1 TO 6:W(X,Y+I)=W(X,Y):NEXT I
800 NEXT X

```

```
810 FOR Y=0 TO 7
820 FOR X=0 TO 133 STEP 7
830 IF W(X,Y)=W(X+7,Y)GOTO 880
840 A1=ABS(W(X,Y)-W(X+7,Y)):A2=INT(A1/7*100+0.5)/100
850 IF W(X,Y)>W(X+7,Y)GOTO 870
860 FOR I=1 TO 6:W(X+I,Y)=W(X,Y)+A2*I:NEXT I:GOTO 890
870 FOR I=1 TO 6:W(X+7-I,Y)=W(X+7,Y)+A2*I:NEXT I:GOTO 890
880 FOR I=1 TO 6:W(X+I,Y)=W(X,Y):NEXT I
890 NEXT X:NEXT Y
900 VTAB 24:HTAB 14:INPUT"N(MAX DATA) = ";N
910 IF Q<19 GOTO 980
920 FOR Y=0 TO 7:FOR X=0 TO 140:FOR I=1 TO N STEP 2
930 IF W(X,Y)<I+0.2 AND W(X,Y)>I-0.2 GOTO 950
940 GOTO 960
950 H PLOT X+10,Y+10+Q*7
960 NEXT I:NEXT X:NEXT Y
970 GOTO 990
980 GOSUB 1130
990 B=Q+1
1000 IF B=20 GOTO 1030
1010 A=B*21
1020 NEXT Q
1030 G=3.14159/180:R=70:HCOLOR=0
1040 FOR C=270 TO 450:V=C*G:X=80+R*COS(V):Y=80-R*SIN(V):H PLOT X,
Y TO X+14,Y:NEXT
1050 FOR C=90 TO 270:V=C*G:X=80+R*COS(V):Y=80-R*SIN(V):H PLOT X, Y
TO 0,Y:NEXT
1060 HCOLOR=7:FOR C=0 TO 360 STEP 0.5:V=C*G:H PLOT 80+R*COS(V), 80-R
*SIN(V):NEXT
1070 H PLOT 78, 154 TO 80, 152 TO 80, 157
1080 PRINT"DO YOU WANT TO SET UP A GRAPHIC FILE?(ANS. Y OR N)": INPUT
G$:D$=CHR$(4)
1090 IF G$="N" GOTO 1120
1100 INPUT"WHAT'S THE GRAPHIC FILE'S NAME?";G$
1110 PRINT D$,"BSAVE",G$,"", A8192, L8191"
1120 END
1130 FOR Y=0 TO 6:FOR X=0 TO 140:FOR I=1 TO N STEP 2
1140 IF W(X,Y)<I+0.2 AND W(X,Y)>I-0.2 GOTO 1160
1150 GOTO 1170
1160 H PLOT X+10,Y+10+Q*7
1170 NEXT I:NEXT X:NEXT Y
```



```
1180 RETURN
1190 INPUT "THE GRAPHIC FILE'S NAME?"; G$
1200 HGR:HCOLOR = 7:HOME
1210 DX = CHR$(4):PRINT D$,"BLOAD";G$
1220 POKE - 16297,0:POKE - 16300,0:POKE - 16302,0:POKE - 16304,0
1230 POKE - 16301,0:VTAB 21
1240 PR 1:PRINT:PRINT TAB(10)"PETROFABRIC DIAGRAM":POKE 1913,65:PRINT
    CHR$(17)
1244 END
1250 GOTO 130
1300 NM = X(1)
1310 FOR I = 2 TO 441
1320 IF NM > X(I) GOTO 1340
1330 NM = X(I)
1340 NEXT I
1350 PRINT "N(MAX DATA) = "; NM: N = NM
```

若展点后即放手让电脑全自动成图可在 490 句后加: GOTO 1300 在 1350 后加: GOTO 500 并将 900 句抹掉即可。